query: PREFIX target: <http://www.wikidata.org/entity/Q84263196> # Replace Q84263196 (COVID-19)with Wikidata ID of your concept of interest, e.g. Q202864 (Zika virus) 
  
  # #defaultView:Graph
  SELECT DISTINCT ?cited_country ?cited_countryLabel ?cited_flag ?citing_country ?citing_countryLabel ?citing_flag
  WITH {
    SELECT DISTINCT ?cited_country ?citing_country (COUNT(?citing_country) AS ?count)
                                    WHERE {
      ?citing_work wdt:P50 ?citing_author . 
      ?citing_work wdt:P921 target: .
      ?cited_work wdt:P921 target: .
      ?citing_work wdt:P2860 ?cited_work . 
      ?cited_work wdt:P50 ?cited_author .  
      FILTER (?citing_work != ?cited_work)
      FILTER NOT EXISTS {
        ?citing_work wdt:P50 ?author .
        ?citing_work wdt:P2860 ?cited_work .
        ?cited_work  wdt:P50 ?author .
      }
      ?citing_author (wdt:P108|wdt:P1416) ?citing_organization . 
      ?cited_author (wdt:P108|wdt:P1416) ?cited_organization . 
      ?cited_organization wdt:P17 ?cited_country.
      ?citing_organization wdt:P17 ?citing_country.
      FILTER (?citing_country != ?cited_country)
    }
    GROUP BY ?cited_country ?citing_country
    ORDER BY DESC(?count)
    LIMIT 30 # Adjust number of connections to display
  } AS %results
  WHERE {
    INCLUDE %results
    ?cited_country wdt:P163 / wdt:P18 ?cited_flag . 
    ?citing_country wdt:P163 / wdt:P18 ?citing_flag . 
   SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }        
   }
  